/*
Landon Schnabel	
Proposal Title: Are Religions Gender-Typed? The Perceived Femininity and Masculinity of Christians, Jews, Muslims, and Atheists

HYPOTHESES

Stated-Hyp1: Christians will be percieved as more feminine and less masculine than
	atheists and people of other religious groups

	Test-Hyp1: Christians women will be percieved as more feminine than non-Christian women

	Test-Hyp2: Christians men will be percieved as less masculine than non-Christian men
	
Stated-Hyp2: Christianity will be percieved as more feminine and less masculine
	than atheism and other religions
	
	Test-Hyp3: Christianity will be percieved as more feminine than atheism and other religions

	Test-Hyp4: Christianity will be percieved as less masculine than atheism and other religions
	
Stated-Hyp3: More religious people will be percieved as more feminine and less 
	masculine among Christians
	
	Test-Hyp5: 	Christian respondents will percieve more religious people as more feminine than less religious people

	Test-Hyp6: 	Christian respondents will percieve more religious people as less masculine than less religious people
	
	
Stated-Hyp4: People who are more religiously dogmatic will be percieved as more 
	masculine and less feminine
	
	Test-Hyp7: 	Religiously dogmatic people will be percieved as less feminine than non-religiously dogmatic people people

	Test-Hyp8: 	Religiously dogmatic people will be percieved as more masculine than non-religiously dogmatic people people
	
Stated-Hyp5: Men whose masculinity is threatened will report lower religiosity and 
	higher religious dogmatism

	Test-Hyp9: 	Among male respondents, a masculinity threat will lead to lower religiousity 
	
	Test-Hyp10: 	Among male respondents, a masculinity threat will lead to more religious dogmatism
	
*/

clear all
use "Schnabel903.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATION

	* vignette 1 (religion and gender)
		/* from quex:

		Condition        Religion    ×    Gender

		Version 1	Christian	Woman
		Version 2	Muslim	Woman
		Version 3	Jewish	Woman
		Version 4	Atheist	Woman
		Version 5	Christian	Man
		Version 6	Muslim	Man
		Version 7	Jewish	Man
		Version 8	Atheist	Man
		*/

		tab P_TESS14_1, mis
		
		
		* gender vignette
		recode P_TESS14_1 (1/4=1) (5/8=0), gen(vig_woman1)
		lab var vig_woman1 "vignette 1 is about woman"
		
		* religion in vignette 1 (version 1 and 5)
		recode P_TESS14_1 (1=1) (5=1) (*=0), gen(vig_christian)
		tab vig_christian
		
	* vignette 2 (religisity+dogmatism+gender)
		/*
		from quex:
		Vignette 2 
		Condition   Religiousness × Dogmatism × Gender
	
		Version 1	Not Religious	Dogmatic	Man
		Version 2	Religious	Dogmatic	Man
		Version 3	Not Religious	Not Dogmatic	Man
		Version 4	Religious	Not Dogmatic	Man
		Version 5	Not Religious	Dogmatic	Woman
		Version 6	Religious	Dogmatic	Woman
		Version 7	Not Religious	Not Dogmatic	Woman
		Version 8	Religious	Not Dogmatic	Woman
		
		NOTE: religion is not specified in vignette 2
		*/

		tab P_TESS14_2, mis
		
		* gender in vignette 2
		recode P_TESS14_2 (1/4=0) (5/8=1), gen(vig_woman2)
		lab var vig_woman2 "vignette 2 is about woman"
		
		* dogmatism in vig 2 (versions 1, 2, 5, 6)
		recode P_TESS14_2 (1/2=1) (3/4=0) (5/6=1) (7/8=0), gen(vig_dogmatic)

		* religiosity in vig 2 (versions 2, 4, 6 ,8)
		recode P_TESS14_2 (1=0) (2=1) (3=0) (4=1) (5=0) (6=1) (7=0) (8=1), ///
		gen(vig_religious)
		
		foreach var of varlist vig* {
		tab `var'
		}
		
	* religion 
		/* from quex:
		Institutions
		Condition                    Institution
		Version 1	Christianity
		Version 2	Judaism
		Version 3	Islam
		Version 4	Atheism
		Version 5	Religion
		Version 6	Spirituality
		*/


		tab P_TESS14_3
		gen christianity=1 if P_TESS14_3==1
		replace christianity=0 if P_TESS14_3>1
		tab christianity, mis

	* threatened masculinity (for H5)
		tab P_TESS14_4
		gen threat_masc=1 if P_TESS14_4==1 & GENDER==1
		replace threat_masc=0 if P_TESS14_4==2 & GENDER==1
		tab threat_masc, mis
	
* CONSTRUCT OUTCOME MEASURES

	* percieved femininity of vignette1
		clonevar perc_fem_vig1 =Q1

	* percieved masculinity of vignette1
		clonevar perc_mas_vig1 =Q2

	* percieved femininity of vignette2
		clonevar perc_fem_vig2 =Q9

	* percieved masculinity of vignette2
		clonevar perc_mas_vig2 =Q10
		
	* percieved masculinity of religion
		clonevar perc_mas_rel =Q7

	* percieved femininity of religion
		clonevar perc_fem_rel =Q6

		// missing values recode
		foreach var of varlist perc* {
		replace `var'=. if `var'>10
		tab `var', mis
		}	
	
* RESPONDENT ATTRIBUTES

	* gender
	tab GENDER
	recode GENDER (1=0) (2=1), gen(female)
	tab female, mis
	
	* R's self-reported religiosity
	tab Q25, mis
		* correct reverse coding
		recode Q25 (4=1) (3=2) (2=3) (1=4) (*=.), gen(r_religiosity)
		lab def r_trait 1 "not at all" 2 "slight" 3 "moderate" 4 "very"
		lab val r_religiosity r_trait
		tab r_religiosity
		lab var r_religiosity "Respondent's self-reported religiosity'"
		
	* R's self-reported dogmatism
	tab Q26, mis
		* correct reverse coding
		recode Q26 (4=1) (3=2) (2=3) (1=4) (*=.), gen(r_dogmatism)
		lab val r_dogmatism r_trait
		tab r_dogmatism
		lab var r_dogmatism "Respondent's self-reported dogmatism"
		
	* R is christian
	tab RELIG
	recode RELIG (1/4=1) (5/11=0) (12=1) (13/14=0) (98=.), gen(r_christian)
	tab r_christian, missing
	
********************************************************************************

* ANALYSIS

* Test-Hyp1: Christian women will be percieved as more feminine than non-Christian women
	reg perc_fem_vig1 i.vig_christian if vig_woman1==1
	// reject. p-val: 0.254
	tess 1.vig_christian +, init(Schnabel903)
	
* Test-Hyp2: Christian men will be percieved as less masculine than non-Christian men	
	reg perc_mas_vig1 i.vig_christian if vig_woman1==0
	// reject. p-val: 0.523
	tess 1.vig_christian -
	
*Test-Hyp3: Christianity will be percieved as more feminine than atheism and other religions
	reg perc_fem_rel i.christianity
	// do not reject. p-val: 0.005 
	tess 1.christianity +, bonf(2)
	
*Test-Hyp4: Christianity will be percieved as less masculine than atheism and other religions
	reg perc_mas_rel i.christianity
	// reject. p-val: 0.973 
	tess 1.christianity -, bonf(2)
	
*Test-Hyp5: Christian respondents will percieve more religious people as more feminine than less religious people
	reg perc_fem_vig2 i.vig_religious if r_christian==1
	// do not reject. 0.571
	tess 1.vig_religious +, bonf(2)

*Test-Hyp6: Christian respondents will percieve more religious people as less masculine than less religious people
	reg perc_mas_vig2 i.vig_religious if r_christian==1
	// reject. 0.689
	tess 1.vig_religious -, bonf(2)	

* Test-Hyp7: Religiously dogmatic people will be percieved as less feminine than non-religiously dogmatic people people

	reg perc_fem_vig2 i.vig_dogmatic if vig_religious==1
	// do not reject. p-val: 0.005
	// this test needs bonferonni correction; test threshold is: 0.05/2
	tess 1.vig_dogmatic -, bonf(2)	
	
* Test-Hyp8: Religiously dogmatic people will be percieved as more masculine than non-religiously dogmatic people people

	reg perc_mas_vig2 i.vig_dogmatic if vig_religious==1	
	// reject. p-val: 0.557
	tess 1.vig_dogmatic +, bonf(2)		
	
* Test-Hyp9: Among male respondents, a masculinity threat will lead to lower religiousity 
	
	reg r_religiosity i.threat_masc if female==0
	// reject. 0.700
	tess 1.threat_masc -, bonf(2)
	
* Test-Hyp10: Among male respondents, a masculinity threat will lead to more religious dogmatism
	
	reg r_dogmatism i.threat_masc if female==0
	// reject. 0.739
	tess 1.threat_masc +, bonf(2)
